Newsgroup: it.comp.www.asp
Titolo: script ASP per contare visite: problema

ciao a tutti, premetto che sono un web designer e non mi intendo
assolutamente di asp. ho un problema con uno script asp per contare le
visite e non posso fare a meno del vostro aiuto!
se metto questo asp (che allego)  nella cartella cgi-bin del mio sito
(hosting ARUBA) e dirigo il browser verso il suo indirizzo (l'ho
chiamato counter.asp)  tutto funziona regolarmente.
ovvero se vado su
http://www.miosito.com/cgi-bin/counter.asp
tutto è perfetto. ad ogni mio passaggio il counter aumenta di uno.
Non funziona invece fuori da cgi-bin e questo è un male: non riesco a
capire come fare in modo di usare questo file in un altra cartella.
Come fare ad esempio se volessi includere questo asp in in un sito che
si trova su un altra cartella, diversa da cgi-bin?
Sul sito 
asp101.com
 dicono di usare questo comando
<!-- #INCLUDE FILE="counter.asp" -->
ma da' errore anche se  lo cambio in
<!-- #INCLUDE FILE="
http://www.bluesigts.com/cgi-bin/counter.asp
" -->
nulla.. non funziona.
Peraltro questo asp  salva il numero dei visitatori in un file di
testo.. cosa buona se volessi inserire il counter nel filmato flash.
---------------------------------------------------------------
<%
'*******************************************************
'*     ASP 101 Sample Code - 
http://www.asp101.com
     *
'*                                                     *
'*   This code is made available as a service to our   *
'*      visitors and is provided strictly for the      *
'*               purpose of illustration.              *
'*                                                     *
'* Please direct all inquiries to 
webm...@asp101.com
 *
'*******************************************************
%>
<%
' Declare our vaiables
Dim objFSO, objCountFile  ' object vars for FSO and File
Dim strCountFileName      ' filename of count text file
Dim iCount                ' count variable
Dim bUseImages            ' boolean whether or not to use images
Dim I                     ' standard looping var
' Determine whether we use images or plain text
' You could just set this to True or False instead
bUseImages = CBool(Request.QueryString("images"))
' Compute our count file's filename
' This is based on the file from which you call count.asp
' It basically takes that name and appends a .cnt so I don't
' accidently overwrite any files.  If for some reason you have
' a file named script_name.asp.cnt then change this or watch out!
strCountFileName =
Server.MapPath(Request.ServerVariables("SCRIPT_NAME") & ".cnt")
' Create FileSystemObject to deal with file access
Set objFSO = Server.CreateObject("Scripting.FileSystemObject")
' Open the file as a text stream (1 = ForReading, True = Create)
Set objCountFile = objFSO.OpenTextFile(strCountFileName, 1, True)
' Read the current count from the file
If Not objCountFile.AtEndOfStream Then
	' Set value to contents of the file
	iCount = CLng(objCountFile.ReadAll)
Else
	' If no file exists or it's empty start at 0
	iCount = 0
End If
' Close the file and destroy the object
objCountFile.Close
Set objCountFile = Nothing
' Increment the count
iCount = iCount + 1
' Overwrite existing file and get a text stream to new one
Set objCountFile = objFSO.CreateTextFile(strCountFileName, True)
' Write updated count
objCountFile.Write iCount
' Close the file and destroy the object
objCountFile.Close
Set objCountFile = Nothing
' Destroy the FSO object
Set objFSO = Nothing
' We're all done with the hard part
' All that's left is to display the results
If bUseImages Then
	' Loop through the count integer showing each digit
	' You can grab the images one at a time or get the zip
	' 
http://www.asp101.com/samples/download/counter_imgs.zip
	For I = 1 to Len(iCount)
		' Output the IMG tag using the right digit
		Response.Write "<IMG SRC=""./images/digit_"
		Response.Write Mid(iCount, I, 1)
		Response.Write ".gif"" ALT="""
		Response.Write Mid(iCount, I, 1)
		Response.Write """ WIDTH=""20"" HEIGHT=""27"">"
	Next 'I
Else
	' No image wanted just show the variable
	Response.Write iCount
End If
%>

 testo.. cosa buona se volessi inserire il counter nel filmato flash.
Prova a dare un percorso assoluto in questo modo:
<!-- #INCLUDE VIRTUAL="/cgi-bin/counter.asp" -->
Ciao, micky